# REPORT DOCUMENTATION PAGE

| | |
|---|---|
| 1a. REPORT SECURITY CLASSIFICATION<br>UNCLASSIFIED | 1b. RESTRICTIVE MARKINGS<br>NONE |
| 2a. SECURITY CLASSIFICATION AUTHORITY | 3. DISTRIBUTION / AVAILABILITY OF REPORT<br>Approved for public release; distribution unlimited. |
| 2b. DECLASSIFICATION / DOWNGRADING SCHEDULE | |
| 4. PERFORMING ORGANIZATION REPORT NUMBER(S)<br>NRL Report 9041 | 5. MONITORING ORGANIZATION REPORT NUMBER(S) |
| 6a. NAME OF PERFORMING ORGANIZATION<br>Naval Research Laboratory<br>6b. OFFICE SYMBOL *(If applicable)*<br>5554 | 7a. NAME OF MONITORING ORGANIZATION |
| 6c. ADDRESS *(City, State, and ZIP Code)*<br>Washington, DC 20375-5000 | 7b. ADDRESS *(City, State, and ZIP Code)* |
| 8a. NAME OF FUNDING / SPONSORING ORGANIZATION<br>Office of Naval Research<br>8b. OFFICE SYMBOL *(If applicable)* | 9. PROCUREMENT INSTRUMENT IDENTIFICATION NUMBER |
| 8c. ADDRESS *(City, State, and ZIP Code)*<br>Arlington, VA 22217 | 10. SOURCE OF FUNDING NUMBERS<br>PROGRAM ELEMENT NO. 62721N<br>PROJECT NO. RX21.221 (STP5110)<br>TASK NO. XF21-221-100<br>WORK UNIT ACCESSION NO. DN 291-241 |

11. TITLE *(Include Security Classification)*
Ground-Wave Path Loss over Ice-Covered Sea at Frequencies Between 0.1 and 10 MHz

12. PERSONAL AUTHOR(S)
Kennedy, Edward J.

| 13a. TYPE OF REPORT | 13b. TIME COVERED | 14. DATE OF REPORT *(Year, Month, Day)* | 15. PAGE COUNT |
|---|---|---|---|
| Interim | FROM 1/86 TO 7/86 | 1987 May 6 | 32 |

16. SUPPLEMENTARY NOTATION

| 17. COSATI CODES | | | 18. SUBJECT TERMS *(Continue on reverse if necessary and identify by block number)* |
|---|---|---|---|
| FIELD | GROUP | SUB-GROUP | Ground wave<br>Propagation over ice<br>Arctic communication |

19. ABSTRACT *(Continue on reverse if necessary and identify by block number)*

The ground-wave attenuation function has been studied for an application involving propagation over an ice-covered sea surface. A review of the techniques that can be used in the analysis of a multilayer ground is presented, and a number of practical sea-ice paths are evaluated numerically. Results are presented for the 0.1 to 10 MHz frequency range and consist of curves showing propagation loss as a function of distance for various ice thicknesses.

It is shown that the variation of the attenuation function with distance and with frequency is significantly different than for the open ocean case. A range of frequencies over which path loss is predicted to be less than for seawater only is dependent on the path length and is shown to lie in the medium frequency (MF) range.

| | | |
|---|---|---|
| 20. DISTRIBUTION / AVAILABILITY OF ABSTRACT<br>☒ UNCLASSIFIED/UNLIMITED ☐ SAME AS RPT. ☐ DTIC USERS | 21. ABSTRACT SECURITY CLASSIFICATION<br>UNCLASSIFIED | |
| 22a. NAME OF RESPONSIBLE INDIVIDUAL<br>Edward J. Kennedy | 22b. TELEPHONE *(Include Area Code)*<br>(202) 767-3677 | 22c. OFFICE SYMBOL<br>Code 5554 |

**DD FORM 1473,** 84 MAR — 83 APR edition may be used until exhausted. All other editions are obsolete.



☆U.S. Government Printing Office: 1985—507-047

Naval Research Laboratory

Washington, DC 20375-5000



NRL Report 9041

# Ground-Wave Path Loss over Ice-Covered Sea at Frequencies Between 0.1 and 10 MHz

EDWARD J. KENNEDY

*Transmission Technology Branch*
*Information Technology Division*

May 6, 1987

Approved for public release; distribution unlimited.

## CONTENTS

I. INTRODUCTION ........ 1

II. GROUND-WAVE ATTENUATION FUNCTION ........ 2

III. NUMERICAL RESULTS ........ 3

IV. SUMMARY ........ 13

V. REFERENCES ........ 14

APPENDIX A — Program Listing ........ 16

APPENDIX B — Useful Approximation for the Ground-Wave Attenuation ........ 25

# GROUND-WAVE PATH LOSS OVER ICE-COVERED SEA AT FREQUENCIES BETWEEN 0.1 AND 10 MHz

## I. INTRODUCTION

Electromagnetic energy may travel between two terminals by any of several propagation mechanisms depending on the distance and nature of the path separating the terminals. Some propagation modes such as meteor scatter are found to vary on a short-term basis while others, including sky-wave ionospheric refraction, may vary in a longer, yet still time-dependent manner.

When the distance between terminals is within or somewhat beyond line-of-sight (LOS), the ground-wave mode may be the predominant propagation mechanism. In general, the ground wave consists of direct and ground-reflected components (existing when the terminals are within LOS), and a surface-wave component that may persist to distances well beyond LOS depending on the frequency and ground characteristics.

The nature of the ground wave has been studied and reported by numerous investigators for a variety of path characteristics including flat and spherical smooth earth. While a number of useful paths have been studied, including seawater and lossy earth, in most cases the electrical parameters of the ground have been assumed to be uniform with depth. Most paths of practical interest including the important case of intratask force communication in the open sea can be analyzed using these models. Barrick [1] provides a useful extension of the theory for a rough sea.

The variation of the ground wave excited over a flat earth surface whose electrical parameters are not uniform with depth was studied by Wait [2] and extended to the spherical surface case [3]. With this analysis, it became possible to predict the important path characteristics for many links of current interest to the Navy, including propagation over an ice-covered sea. In papers by Hill and Wait [4,5], it was shown that at high frequency (HF), the effect of the ice layer is to produce an enhanced field strength (over the seawater only case) at shorter ranges, but with serious degradation as the distance is increased beyond a point that is strongly dependent on frequency.

Whether propagation occurs over homogeneous or stratified earth, the ground-wave mode exhibits a number of important advantages over other propagation types. These include:

- path loss is not time dependent over intervals associated with typical message traffic;
- optimum frequency can be predicted in advance of communication sessions and does not require active channel probing as may be the case for sky-wave or meteor-scatter modes;
- frequency dispersion is insignificant (although for propagation over sea ice, this may be a factor at some ranges and frequencies);
- unaffected by natural or man-made ionospheric disturbances, and

---

Manuscript approved December 11, 1986.

- there is a built-in advantage over interceptors and jammers located beyond the communication range because of the sharp increase in attenuation at such distances.

The purpose of this report is to extend the work presented by previous investigators to include the low-frequency (LF) and medium-frequency (MF) range beginning at 0.1 MHz where the electrical properties of an ice-covered sea will be shown to produce enhanced link performance over distances of practical importance.

## II. GROUND-WAVE ATTENUATION FUNCTION

Under unrestricted conditions and following the general analysis of Hill and Wait [6], the electric field strength, $E$, produced at a distance from a vertical electric dipole, is given by:

$$E = E_0 W, \tag{1}$$

where $E_0$ is the field strength produced by the same source over a plane, perfectly conducting surface and $W$ is the attenuation function that includes effects caused by the spheroidal and stratified surface shown in Fig. 1. Medium 1 is the layer having a conductivity $\sigma_1$ and dielectric constant $\epsilon_1$. The lower region, assumed to extend to great depth, is characterized by conductivity $\sigma_2$ and dielectric constant $\epsilon_2$. The upper half-space is assumed to be equivalent to free space with no conductivity and dielectric constant $\epsilon_0$. The permeability in all regions is equal to $\mu_0$.




Figure 1 — Propagation over stratified, spherical earth

When the height of the dipole source and the point at which the field strength is measured are both at the upper surface of medium 1, $W$ can be written as the residue series:

$$W = \sqrt{\frac{\pi x}{i}} \sum_{s=1}^{\infty} \frac{e^{(-ixt_s)}}{(t_s - q^2)}, \tag{2}$$

where $x = \left(\frac{ka}{2}\right)^{1/3} \left(\frac{d}{a}\right)$ is the numerical distance, $q = -i\left(\frac{ka}{2}\right)^{1/3} \Delta$, $\Delta = \frac{Z}{120\pi}$, $k = \omega\sqrt{\epsilon_0\mu_0}$, $a$ is the radius of the earth, and $\omega = 2\pi f$ is the angular frequency.

The surface impedance $Z$ at the boundary between medium 1 and the upper half-space is given [4] by:

$$Z \cong K_1 \frac{K_2 + K_1 \tanh u_1 h_1}{K_1 + K_2 \tanh u_1 h_1}, \tag{3}$$

where

$$K_1 = \frac{u_1}{\sigma_1 + i\epsilon_1\omega}, \qquad u_1 = \sqrt{\gamma_1^2 + \lambda_s^2},$$

$$K_2 = \frac{u_2}{\sigma_2 + i\epsilon_2\omega}, \qquad u_2 = \sqrt{\gamma_2^2 + \lambda_s^2},$$

$$\gamma_1^2 = i\mu_0\omega(\sigma_1 + i\epsilon_1\omega), \qquad \gamma_2^2 = i\mu_0\omega(\sigma_2 + i\epsilon_2\omega), \text{ and}$$

$$\lambda_s = k + \left(\frac{ka}{2}\right)^{1/3} \left(\frac{t_s}{a}\right).$$

The values $t_s$, which make up the poles of Eq. (2) and appear in the definition for $\lambda_s$, are roots of the equation,

$$\frac{dw(t)}{dt} - qw(t) = 0, \tag{4}$$

where

$$w(t) = \sqrt{\pi}\{Bi(t) - iAi(t)\}.$$

The symbols $Ai(t)$ and $Bi(t)$ refer to Airy functions [7]. The system of Eqs. (3) and (4) can be solved iteratively for the roots $t_s$ (the program listing given in Appendix A provides this procedure). However, as pointed out by Hill and Wait [4], for the case of Arctic ice overlaying a highly conductive sea, a reasonable approximation for $Z$ in Eq. (3) can be obtained by setting $\lambda_s = 0$.

A general procedure for solving Eq. (4) for the roots $t_s$ is given in Hill and Wait [6] and is briefly summarized in Appendix B. Two simpler forms for the ground-wave attenuation function $W$ can be used at relatively short numerical distances $x$. The first of these, called the *small curvature expansion*, is a modification to the flat earth attenuation function for a spherical surface and is useful for large $|q|$. The second form, called the *power series representation*, is most useful at short ranges and small $|q|$ (e.g., a nonlayered, highly conducting surface such as open ocean). Both of these approximations along with their regions of validity are presented in Appendix B.

## III. NUMERICAL RESULTS

The equations for the ground-wave attenuation function have been solved for the case of sea ice of various thicknesses overlaying seawater. In the results that follow, a dielectric constant $\epsilon_{r2} = 80$ and conductivity $\sigma_2 = 4$ Siemans was assumed for the seawater constituting medium 2. (A description of a special, multilayer case follows in which a lower conductivity seawater layer exists just below the ice sheet as a result of melting.) For the sea-ice layer, the conductivity and dielectric constant are known to be dependent on frequency, temperature, and the age of the ice with cold or multiyear ice being much less conductive than warm or first-year ice. Since the variation of conductivity with frequency is less pronounced, a constant value for $\sigma_1 = 0.000333$ Siemans was used for the sea-ice layer of medium 1. This corresponds to one-year-old ice at $-10°$C as given by Wentworth and Cohn [8]. The variation of dielectric constant $\epsilon_{r1}$ with frequency is more strongly dependent on frequency and is also derived from data given in Ref. 8 (see Fig. 2). Another useful quantity that can be used to describe the electrical quality of the ice sheet is the loss tangent defined as $\tan\delta = \sigma/\omega\epsilon_r\epsilon_0$ [9].

One of the major difficulties in analyzing a sea-ice path is the considerable range of electrical parameters that is possible for the ice layer depending on season and geographic location. It is important to note that the results presented here are based on the particular values chosen for the dielectric constant and conductivity of the ice layer. Although they are believed to be representative of a typical Arctic path, some variation in link characteristics can be expected depending on the specific geography chosen.

The total path loss is presented in the results that follow, including effects due to the attenuation function $W$ as well as the distance dependence of the term $E_0$ in Eq. (1). Furthermore, it is assumed that for any given case, the electrical properties of the surface over the complete path are constant and that the thickness of the layer is uniform. Although techniques are available for analysis of a mixed path in which the ice layer varies in its electrical properties and thickness, this has not been considered here.




Figure 2 — Relative dielectric constant variation as a function of frequency used in calculations of ground-wave path loss

Figures 3 to 12 show the numerical results of the path attenuation in decibels as a function of distance in kilometers for frequencies between 0.1 and 10 MHz. Table 1 provides a summary of the data presented in each figure as well as the dielectric constant and loss tangent used in the calculation. Each figure also includes two curves for a uniform earth (medium 1 = medium 2), one for seawater only and one for sea ice only, for comparison with the layered cases. Ice thicknesses up to 7 m were chosen for study since thicker layers are rarely encountered over a large horizontal scale. A typical thickness of 3 m is frequently given for the central Arctic. Only thinner ice sheets are considered at the higher frequencies because of the much stronger effect on the path attenuation.




Figure 3 — Ground-wave attenuation as a function of distance over smooth spherical earth for uniform ice and seawater surfaces and for three layered cases. The frequency is 100 kHz.




Figure 4 — Ground-wave attenuation as a function of distance over smooth spherical earth for uniform ice and seawater surfaces and for three layered cases. The frequency is 200 kHz.




Figure 5 — Ground-wave attenuation as a function of distance over smooth spherical earth for uniform ice and seawater surfaces and for three layered cases. The frequency is 300 kHz.




Figure 6 — Ground-wave attenuation as a function of distance over smooth spherical earth for uniform ice and seawater surfaces and for two layered cases. The frequency is 500 kHz.




Figure 7 — Ground-wave attenuation as a function of distance over smooth spherical earth for uniform ice and seawater surfaces and for three layered cases. The frequency is 700 kHz.




Figure 8 — Ground-wave attenuation as a function of distance over smooth spherical earth for uniform ice and seawater surfaces and for three layered cases. The frequency is 1 MHz.




Figure 9 — Ground-wave attenuation as a function of distance over smooth spherical earth for uniform ice and seawater surfaces and for two layered cases. The frequency is 3 MHz.




Figure 10 — Ground-wave attenuation as a function of distance over smooth spherical earth for uniform ice and seawater surfaces and for two layered cases. The frequency is 5 MHz.




Figure 11 — Ground-wave attenuation as a function of distance over smooth spherical earth for uniform ice and seawater surfaces and for three layered cases. The frequency is 7 MHz.




Figure 12 — Ground-wave attenuation as a function of distance over smooth spherical earth for uniform ice and seawater surfaces and for three layered cases. The frequency is 10 MHz.

Table 1 — Data Summary

| Figure | Frequency (MHz) | Ice Thickness (m) | Relative Dielectric Constant | Tan δ |
|---|---|---|---|---|
| 3 | 0.1 | 3,5,7 | 22 | 1.2 |
| 4 | 0.2 | 3,5,7 | 16 | 1.9 |
| 5 | 0.3 | 3,5,7 | 13 | 1.3 |
| 6 | 0.5 | 3,5 | 10 | 1.0 |
| 7 | 0.7 | 3,5,7 | 8 | 1.0 |
| 8 | 1.0 | 1,3,5 | 7 | 1.0 |
| 9 | 3.0 | 1,3 | 6 | 0.333 |
| 10 | 5.0 | 1,3 | 6 | 0.200 |
| 11 | 7.0 | .5,1,3 | 6 | 0.143 |
| 12 | 10.0 | .5,1,2 | 6 | 0.100 |

In the calculation of these results, one of the three techniques described in Appendix B was used. In general, for a given case, either the power series or small curvature approximation was used at the short range and the residue series for the longer ranges. To obtain a smooth fit at the transition range, 50 terms were used in the residue series summation.

The general characteristic of the variation of path loss for the layered case can be described in terms of three regions. The lower frequency curves (Figs. 3 to 6) show only the first of these zones, a region at short ranges where the surface wave mode dominates and the field strength is enhanced over the seawater only case. The effect is minor for the thicknesses studied (and of practical interest) at the lowest frequencies but is significant at 300 and 500 kHz where the path loss is as much as 25 dB better than seawater alone at a range of 1000 km (546 nmi). The region of enhanced propagation is evident in all higher frequency curves although at distances that become shorter as the frequency increases. At the two highest frequencies studied, 5 and 7 MHz, the enhanced zone still exists but at distances too short to be useful for most communication purposes.

The second, intermediate zone is a region in which the field strength decreases rapidly and non-monotonically with distance. The great increase in path loss over this interval is caused by the decay of the surface-wave component of the total field in this region. At some distance beyond the onset of this rapid increase in path loss, the surface-wave mode decays to a magnitude that is comparable to the other normal ground-wave modes, and an interference region consisting of maxima and minima in the attenuation function is evident. Although not shown here, there is a corresponding oscillatory effect in the phase of the electric field as a function of range [5] rendering this portion of the path dispersive.

Because the surface-wave mode has a greater exponential attenuation than the other modes making up the total ground-wave field, its effect on the behavior of the total path loss is no longer important beyond the intermediate region. In the third zone, the attenuation function returns to a smooth, monotonically decreasing function of range but at a much greater loss than for seawater alone, in spite of the existence of a seawater substrate. Moreover, in this region the shape of the layered case curves matches the ice-only curve, indicating that the ground wave in this zone is dominated by the ice and not by the seawater substrate. It is readily apparent that at frequencies above approximately 2 to 3 MHz for ice thicknesses likely to be encountered over deep Arctic ground-wave paths, the path loss is much greater than for an open ocean geometry, limiting the range of available frequencies for communication purposes to the MF and lower HF ranges.

This effect is seen more graphically in Fig. 13 where the path loss is plotted as a function of frequency for an ice thickness of 3 m for four distances, 50, 200, 800, and 1600 km. For comparison, the dashed curve shows the attenuation as a function of frequency for a smooth seawater-only case at a range of 50 km. (In the solid curves, the oscillatory behavior associated with the layered cases at intermediate ranges has been smoothed for clarity.) Several interesting characteristics are evident in Fig. 13 including:

- For any given range, at the lowest frequencies, the layered and uniform earth curves are asymptotic. Apparently the layer thickness (and its effect) is trivial at these frequencies in comparison with the wavelength of the electric field in air.

- There is a frequency or set of frequencies where the layered earth propagation enhancement over uniform seawater is optimized. For a distance of 50 km, this frequency is near 1 MHz while at the much longer distance, 1600 km, the optimum frequency has dropped to approximately 500 kHz.

- The expected increasing trend of attenuation with frequency that is evident in the uniform seawater and short-range layered cases is reversed for the two longer distance cases plotted. At 1600 km, for example, frequencies near 500 kHz have less path loss than lower frequencies in the range 100 to 300 kHz.




Figure 13 — Ground-wave attenuation over spherical earth as a function of frequency

- Each of the solid curves contains a sharp knee in its variation with frequency. This feature becomes more prominent as the distance increases. At 1600 km, for example, the difference in path loss between 600 and 800 kHz is on the order of 40 dB. The effect is similar to waveguide cutoff and highlights the importance of careful choice of frequency in a practical case.

- Frequencies in the HF band (3 to 30 MHz) are of value only for short distance links via ground-wave propagation. Longer distance links are better addressed using frequencies in the MF range (0.3 to 3 MHz). The availability of communication hardware capable of matching this need, however, is limited.

Figure 14 shows the variation of path loss as a function of layer depth for 10 and 50 km at a fixed frequency of 7 MHz. The two curves approach the seawater-only case (shown as horizontal lines at the ordinate axis) as the thickness is reduced to zero. At the right edge of the figure, the curves asymptotically approach the case of propagation over uniform sea ice as the thickness increases indefinitely. For small ice depths, a modest enhancement over uniform seawater is shown for both curves. As the thickness increases, the path loss increases rapidly reaching a maximum attenuation at a thickness of approximately 4.745 m and then reversing this trend to reach a minimum near 9.44 m. A succession of maxima and minima follow the envelopes shown for further thickness increases. This effect is a consequence of the impedance transformation effect due to the ice layer and described in Eq. (3). The underlying highly conductive sea substrate is alternately transformed by this mechanism in a manner similar to a conventional transmission line to produce a surface impedance that is, alternately, a good conductor and a poor conductor. Since the ice is not a perfect dielectric and possesses some loss, successive maxima and minima do not reach preceding levels and the underlying sea is eventually masked by ice losses as the thickness continues to increase.

Figure 15 shows three cases of a multilayer geometry at 700 kHz for comparison with a uniform 3-m thick sea-ice layer. In the curve identified as case A, six layers of sea ice, each 0.5-m thick are used in the computer model to form a piecewise approximation to the variation of conductivity with depth in a 3-m thick sheet. Table 2 gives the values used that were obtained from Wentworth and




Figure 14 — Ground-wave attenuation at fixed distances of 10 and 50 km as a function of ice layer thickness at 7 MHz




Figure 15 — Ground-wave attenuation as a function of distance for propagation over an ice covered sea surface where the electrical properties of the layer vary with depth

Table 2 — Ice Electrical Properties Used in Fig. 15

| Depth (m) | Curve A | | Curve B | | Curve C | |
|---|---|---|---|---|---|---|
| | $\epsilon_r$ | $\sigma$ | $\epsilon_r$ | $\sigma$ | $\epsilon_r$ | $\sigma$ |
| 0.5 | 5.5 | 0.000100 | 5.5 | 0.000154 | 5.5 | 0.000100 |
| 1.0 | 6.5 | 0.000159 | 6.0 | 0.000210 | 6.5 | 0.000159 |
| 1.5 | 8.0 | 0.000251 | 6.0 | 0.000280 | 8.0 | 0.000251 |
| 2.0 | 10.0 | 0.000398 | 6.5 | 0.000341 | 10.0 | 0.000398 |
| 2.5 | 12.0 | 0.000631 | 7.0 | 0.000423 | 12.0 | 0.000631 |
| 3.0 | 18.0 | 0.001000 | 7.5 | 0.000467 | 18.0 | 0.001000 |
| 3.5 | 80.0 | 1.000000 | 80.0 | 4.000000 | 80.0 | 4.000000 |
| >4.0 | 80.0 | 4.000000 | 80.0 | 4.000000 | 80.0 | 4.000000 |

Cohn [10]. In certain seasons, ice melt can produce a layer of fresh water that tends to stratify at the ice-water boundary since it is less dense than regular seawater. For case A, this fresh-water layer has been included in the model with a thickness of 0.5 m and a conductivity $\sigma = 1$ Sieman. Curve C is identical in its parameters to Curve A with the exception that the layer of fresh water is not present. Curve B contains data obtained from a core sample in the marginal ice zone during an Arctic field test.

Differences in the curves are not significant until the distance increases to beyond approximately 200 km. At the greatest range shown, the maximum difference between curves is on the order of 10 dB. Curves A and C are plotted for ice parameters that are significantly less lossy than for the ice used for Curve B, but Curve A shows significantly more attenuation at the greatest range because of the less conductive seawater substrate. Note that curves B and C, which describe a more realistic ice layer, are still within 3 dB of the uniform parameter 3-m case even at the longest range considered.

## IV. SUMMARY

The ground-wave propagation mode has been studied for applications involving a sea-ice path. In general, the advantages of this propagation mechanism are:

- The path loss is not time-dependent over intervals associated with normal communication message length. The optimum frequency can be predicted in advance of communication sessions and does not require active channel probing as may be the case for skywave or meteor scatter propagation.

- Frequency dispersion is insignificant (although for propagation over ice, this may not be true at some ranges and frequencies).

- There is a large margin against jammers or interceptors who are located beyond the communication range.

- Ground-wave propagation is unaffected by natural or man-made ionospheric disturbances and, in fact, may be enhanced during such events since external noise and interference levels can be expected to diminish in response to ionospheric failure.

In the Arctic, some of these advantages are magnified. The normal knee in the attenuation curve for seawater is changed appreciably for an ice-covered path as a result of the rapid decay of the surface wave component of the total field. Correct choice of operating frequency can simultaneously optimize communication link performance while discriminating against potential interceptors or jammers.

It has been shown that the path characteristics are sensitive to the electrical properties of the surface, particularly the ice thickness, but also the conductivity and dielectric constant of the ice. During those times of the year when the ice melts, path loss will probably be affected by the presence of a fresh water layer.

The general behavior of the attenuation as a function of range is given to a good degree of accuracy by considering the layer to be of uniform electrical characteristics throughout its thickness. An analysis of the effect of changes in layer thickness over the path length has not been done. It is expected that the higher frequencies will be most strongly affected by ice thickness variability since small changes will represent a significant portion of a wavelength.

In the Arctic, the best frequencies for use in ground-wave links range from 100 kHz or below to approximately 2 MHz depending on path length. When the attenuation characteristics are combined with other system parameters such as antenna efficiency and external noise level, the optimum communication frequency is expected to be readily identifiable as a peak in system performance over a small range of frequencies in the MF band.

Complete system analysis must also consider the effects of ionospheric refraction on link performance. Electromagnetic energy reaching a distant terminal by this mode may be of comparable magnitude to the ground wave leading to a zone of interference between the two components. In addition, the Artic ionosphere is known to be highly variable with frequent periods of ionospheric failure. Under these conditions, external noise and interfering signals that reach the Arctic by sky wave modes can be expected to decrease markedly leading to improved link performance.

## V. REFERENCES


1. D.E. Barrick, ''Theory of HF and VHF Propagation Across the Rough Sea, 2, Applications to HF and VHF Propagation Above the Sea,'' *Radio Sci.* **6**(5), 527 (1971).

2. J.R. Wait, ''Radiation from a Vertical Dipole over a (Plane) Stratified Ground,'' *Trans. IRE* **AP-1**, 9 (1953).

3. J.R. Wait, ''Radiation from a Vertical Antenna over a Curved Stratified Ground,'' *J. Res. Nat. Bur. Standards* **56**(4), 237 (1956).

4. D.A. Hill and J.R. Wait, ''HF Ground Wave Propagation over Sea Ice for a Spherical Earth Model,'' *IEEE Trans. Antennas Propag.*, **AP-29**(3), 525 (1981).

5. D.A. Hill and J.R. Wait, ''HF Radio Transmission over Sea Ice and Remote Sensing Possibilities,'' *IEEE Trans. Geos. Rem. Sens.* **GE-19**(4), 204 (1981).

6. D.A. Hill and J.R. Wait, ''Ground Wave Attenuation Function for a Spherical Earth with Arbitrary Surface Impedance,'' *Radio Sci.* **15**(3) (1980).

7. M. Abramowitz and I.A. Stegun, *Handbook of Mathematical Functions*, National Bureau of Standards (Applied Mathematics Series 55, Nov. 1970).

8. F.L. Wentworth and M. Cohn, ''Electrical Properties of Sea Ice at 0.1 to 30 Mc/s,'' *Radio Sci., J. Res. NBS* **68D**, 681 (1964).

9. S. Ramo, J.R. Whinnery, and T. Van Duzer, *Fields and Waves In Communication Electronics* (New York, John Wiley & Sons, 1967).

# Appendix A
# PROGRAM LISTING

```
10   OVERLAP
20   RAD
30   COM W,E0,U0,Er(10),Si(10),Hi(10),Gr(10),Gi(10),Kfs,Re,Nrt,Nla
40   COM Tsr(101),Tsi(101),Acr(10),Aci(10),Qr(101),Qi(101),Zr(10),Zi(10)
50   DIM Yi1r(100),Yi1i(100),Yi2r(100),Yi2i(100)
60   Nrt=100
70   DISP "                WAIT - MATRIX INITIALIZATION"
80   PRINTER IS 16
90   CALL Cexp(0,-PI/3,Rr,Ri)
100  DATA 1.01879297,3.24819758,4.82009921,6.16330736,7.37217726,8.48848673
110  DATA 9.53544905,10.52766040,11.47505663,12.38478837
120  RESTORE 100
130  FOR J=1 TO 10
140  READ X5
150  Yi1r(J)=X5*Rr
160  Yi1i(J)=X5*Ri
170  NEXT J
180  DATA 2.33810741,4.08794944,5.52055983,6.78670809,7.94413359,9.02265085
190  DATA 10.04017434,11.00852430,11.93601556,12.82877675
200  RESTORE 180
210  FOR J=1 TO 10
220  READ X5
230  Yi2r(J)=X5*Rr
240  Yi2i(J)=X5*Ri
250  NEXT J
260  FOR J=11 TO Nrt
270  Ais=3*PI*(4*J-1)/8
280  Aisp=3*PI*(4*J-3)/8
290  CALL Azero(Ais,X5)
300  CALL Azerop(Aisp,Y5)
310  Yi1r(J)=Y5*Rr
320  Yi1i(J)=Y5*Ri
330  Yi2r(J)=X5*Rr
340  Yi2i(J)=X5*Ri
350  NEXT J
360  MAT Tsr=ZER
370  MAT Tsi=ZER
380  MAT Qr=ZER
390  MAT Qi=ZER
400  MAT Er=CON
410  MAT Si=ZER
420  MAT Hi=ZER
430  N0=120*PI
440  MAT Zr=(N0)
450  MAT Zi=ZER
460  MAT Gr=ZER
470  MAT Gi=ZER
480  Nrt=50
490  PRINT PAGE
500  INPUT "ENTER NUMBER OF ROOTS TO BE FOUND (50)",Nrt
510  Xdd=.2
520  Fn=Qqr=1
530  Qqi=Fc=Fss=0
540  E0=8.854E-12
550  Re=6378000*4/3
560  U0=PI*4E-7
570  C=1/SQR(E0*U0)
580  INPUT "ENTER FREQUENCY (MHz) ",F
590  W=PI*F*2E6
600  Lfs=C/(F*1E6)
610  Kfs=2*PI/Lfs
620  FOR L=1 TO 10
630  DISP "ENTER DIELECTRIC CONSTANT OF LAYER ";L;" (-1 TO STOP)";
640  INPUT Er(L)
650  IF Er(L)<>-1 THEN 680
660  IF L=1 THEN 630
670  GOTO 730
680  DISP "ENTER CONDUCTIVITY OF LAYER ";L;
690  INPUT Si(L)
700  DISP "ENTER THICKNESS OF LAYER ";L;
```

```
710  INPUT Hi(L)
720  NEXT L
730  P1$="Y"
740  INPUT "DO YOU WANT HARD COPY? (Y/N)",P1$
750  IF P1$="Y" THEN PRINTER IS 7,6,WIDTH(132)
760  Nla=L-1
770  PRINT "LAYER # ";0;": Eps=";1;TAB(30);"CONDUCTIVITY=";0
780  FOR L=1 TO Nla-1
790  PRINT "LAYER # ";L;": Eps=";Er(L);TAB(30);"CONDUCTIVITY=";Si(L);TAB(60);"THICKNESS=";Hi(L);"meters"
800  NEXT L
810  PRINT "LAYER # ";Nla;": Eps=";Er(Nla);TAB(30);"CONDUCTIVITY=";Si(Nla)
820  A1=(Kfs*Re/2)^(1/3)
830  DISP "            WAIT - SURFACE IMPEDANCE CALCULATIONS"
840  FOR L=1 TO Nla
850  Gr(L)=-Er(L)*E0*W*W*U0
860  Gi(L)=Si(L)*W*U0
870  PRINT "Gamma ";L,Gr(L),Gi(L)
880  NEXT L
890  Niter=-1
900  FOR N=1 TO Nrt+1
910  CALL Zsurf(Tsr(N),Tsi(N),Zr,Zi)
920  Der=Zr/N0
930  Dei=Zi/N0
940  Qr(N)=A1*Dei
950  Qi(N)=-A1*Der
960  IF Qr(1)>0 THEN 1000
970  Fss=1
980  Xdd=1000
990  GOTO 1010
1000 NEXT N
1010 Niter=Niter+1
1020 Qr=Qr(1)
1030 Qi=Qi(1)
1040 Qmag=SQR(Qr*Qr+Qi*Qi)
1050 Argq=180/PI*ATN(Qi/Qr)
1060 Fn=1
1070 IF Qmag<=1 THEN 1100
1080 Fn=2
1090 CALL Cdivid(1,0,Qr,Qi,Qqr,Qqi)
1100 PRINT "F=";F;" MHz","Qr=";Qr,"Qi=";Qi
1110 PRINT CHR$(124)&"Q"&CHR$(124)&"=";SQR(Qr*Qr+Qi*Qi);TAB(30);"(Q=";Argq,LIN(1)
1120 PRINT CHR$(124)&"D"&CHR$(124)&"=";SQR(Der*Der+Dei*Dei);TAB(30);"(D=";180/PI*ATN(Dei/Der)
1130 PRINT LIN(1);"NUMBER OF ITERATIONS = ";Niter;LIN(1)
1140 IF Fss=1 THEN 1450
1150 Jadd=Fc=0
1160 IF Argq<-30 THEN 1300
1170 IF Argq>-19.2 THEN 1200
1180 Fn=1
1190 GOTO 1300
1200 IF Qmag<=1 THEN 1300
1210 IF Qmag>5 THEN 1270
1220 IF (Qmag>2.5) AND (Argq>-.01) THEN 1270
1230 CALL Kutta(0,0,200,Qr,Qi,Yi1r(1),Yi1i(1),Tsr(1),Tsi(1),1)
1240 CALL Rt0(Qr,Qi,Atr,Ati)
1250 PRINT "Tsr(g)=";Atr,"Tsi(g)=";Ati
1260 GOTO 1280
1270 CALL Rt0(Qr,Qi,Tsr(1),Tsi(1))
1280 PRINT USING 1390;"Tsr(",1,")=",Tsr(1),"Tsi(",1,")=",Tsi(1),"Qr=",Qr(1),"Qi=",Qi(1)
1290 Jadd=1
1300 FOR J=1 TO Nrt
1310 Js=J+Jadd
1320 DISP "            ITERATION #";Niter;"  ROOT #";Js
1330 IF Fn=2 THEN 1360
1340 CALL Kutta(0,0,50,Qr(Js),Qi(Js),Yi1r(J),Yi1i(J),Tsr(Js),Tsi(Js),1)
1350 GOTO 1380
1360 CALL Cdivid(1,0,Qr(Js),Qi(Js),Qqr,Qqi)
1370 CALL Kutta(0,0,50,Qqr,Qqi,Yi2r(J),Yi2i(J),Tsr(Js),Tsi(Js),2)
1380 PRINT USING 1390;"Tsr(",Js,")=",Tsr(Js),"Tsi(",Js,")=",Tsi(Js),"Qr=",Qr(Js),"Qi=",Qi(Js)
1390 IMAGE 2(4X,4A,DD,2A,DDDD.DDDD,2X),3X,2(4X,3A,MDDD.DDDD)
1400 NEXT J
1410 BEEP
1420 Q$="N"
1430 INPUT "Iterate Roots Again (Y/N)?",Q$
1440 IF Q$="Y" THEN 900
1450 INPUT "ENTER DISTANCES: Start, Multiplier, Number (in km)",Din,Dmult,Ndi
1460 Dd=Din*1000
1470 Ns=Nrt+Jadd
1480 W1=20*LGT(Lfs/(2*PI))
1490 Wm=1
1500 IF Qmag>1 THEN 1520
1510 CALL Ac(Qr,Qi)
1520 FOR Kin=1 TO Ndi
1530 X=A1*Dd/Re
1540 IF X>Xdd THEN 1660
1550 IF Qmag<1 THEN 1620
1560 CALL Scurv(X,Qr,Qi,Wr,Wi,Rns,Nfp,Mer)
1570 IF Nfp=2 THEN 1600
```

```
1580 N$="Small Curvature-1"
1590 GOTO 1680
1600 N$="Small Curvature-2"
1610 GOTO 1680
1620 CALL Pseries(X,Qr,Qi,Wr,Wi)
1630 N$="Power Series"
1640 Rns=10
1650 GOTO 1680
1660 CALL Rseries(X,Ns,Wr,Wi,Rns,Mer)
1670 N$="Residue Series"
1680 CALL Cabs(Wr,Wi,Wm)
1690 Wpl=W1+20*LGT(1/Dd)+20*LGT(Wm)
1700 PRINT USING 1710;Dd/1000,Wpl,N$,"#TERMS= ",Rns,"Error=",Mer
1710 IMAGE 4X,DDDD.DDD,9X,SDDD.D,10X,18A,5X,8A,DD,4X,6A,D.DE
1720 Dd=Dd*Dmult
1730 NEXT Kin
1740 Q$="N"
1750 INPUT "New Distances (Y/N)?",Q$
1760 IF Q$="Y" THEN 1450
1770 Q$="N"
1780 INPUT "Iterate Roots Again (Y/N)?",Q$
1790 IF Q$="Y" THEN 900
1800 END !

1810 DEF FNTanh(X)
1820 IF ABS(X)<1 THEN 1870
1830 IF ABS(X)>10 THEN 1900
1840 Ex=EXP(X)
1850 Tanh=(Ex-1/Ex)/(Ex+1/Ex)
1860 RETURN Tanh
1870 D2x=EXP(2*X)-1
1880 Tanh=D2x/(2+D2x)
1890 RETURN Tanh
1900 Tanh=1
1910 RETURN Tanh
1920 FNEND !

1930 SUB Cdivid(A1,B1,A2,B2,R,I)
```

```
1940 C=A2*A2+B2*B2
1950 IF C<>0 THEN 2000
1960 PRINT LIN(2),"ERROR IN SUBPROGRAM Cdivid."
1970 PRINT "DIVISOR IS ZERO.",LIN(2)
1980 PAUSE
1990 SUBEXIT
2000 R=(A2*A1+B2*B1)/C
2010 I=(A2*B1-B2*A1)/C
2020 SUBEND !

2030 SUB Cexp(A,B,R,I)
2040 R=EXP(A)*COS(B)
2050 I=EXP(A)*SIN(B)
2060 SUBEND !

2070 SUB Cabs(X,Y,Cabs)
2080 X1=ABS(X)
2090 Y1=ABS(Y)
2100 IF X1<>0 THEN 2130
2110 Cabs=Y1
2120 SUBEXIT
2130 IF Y1<>0 THEN 2160
2140 Cabs=X1
2150 SUBEXIT
2160 IF X1>Y1 THEN Cabs=X1*SQR(1+(Y1/X1)^2)
2170 IF X1<=Y1 THEN Cabs=Y1*SQR(1+(X1/Y1)^2)
2180 SUBEND !

2190 SUB Csqrt(A,B,R,I)
2200 IF (A<>0) OR (B<>0) THEN 2230
2210 R=I=0
2220 SUBEXIT
2230 CALL Cabs(A,B,Cabs)
2240 R=SQR((ABS(A)+Cabs)*.5)
2250 IF A<0 THEN 2280
2260 I=B/(R+R)
2270 SUBEXIT
2280 IF B<0 THEN I=-R
2290 IF B>=0 THEN I=R
2300 R=B/(I+I)
2310 SUBEND !

2320 SUB Zsurf(Tsr,Tsi,Zr,Zi)
2330 COM W,E0,U0,Er(10),Si(10),Hi(10),Gr(10),Gi(10),Kfs,Re,Nrt,Nla
2340 COM Tsr(101),Tsi(101),Acr(10),Aci(10),Qr(101),Qi(101),Zr(10),Zi(10)
2350 A1=(Kfs*Re/2)^(1/3)/Re
2360 Lsr=Kfs+A1*Tsr
2370 Lsi=A1*Tsi
2380 Lsr2=Lsr*Lsr-Lsi*Lsi
2390 Lsi2=2*Lsr*Lsi
2400 FOR M=Nla TO 2 STEP -1
2410 CALL Csqrt(Gr(M-1)+Lsr2,Gi(M-1)+Lsi2,Uir,Uii)
2420 CALL Csqrt(Gr(M)+Lsr2,Gi(M)+Lsi2,Uwr,Uwi)
2430 CALL Cdivid(Uir,Uii,Si(M-1),E0*Er(M-1)*W,Kir,Kii)
2440 IF M=Nla THEN 2480
2450 Kwr=Zr(M)
2460 Kwi=Zi(M)
2470 GOTO 2490
2480 CALL Cdivid(Uwr,Uwi,Si(M),E0*Er(M)*W,Kwr,Kwi)
2490 Ar=FNTanh(Uir*Hi(M-1))
2500 Ai=TAN(Uii*Hi(M-1))
2510 CALL Cdivid(Ar,Ai,1,Ar*Ai,Tr,Ti)
2520 Nr=Kir*Tr-Kii*Ti+Kwr
2530 Ni=Kir*Ti+Kii*Tr+Kwi
2540 Dr=Kwr*Tr-Kwi*Ti+Kir
2550 Di=Kwr*Ti+Kwi*Tr+Kii
```

```
2560 CALL Cdivid(Nr,Ni,Dr,Di,Ar,Ai)
2570 Zr(M-1)=Kir*Ar-Kii*Ai
2580 Zi(M-1)=Kir*Ai+Kii*Ar
2590 NEXT M
2600 Zr=Zr(1)
2610 Zi=Zi(1)
2620 SUBEND !

2630 SUB Rseries(X,Ns,Wr,Wi,Rns,Mer)
2640 COM W,E0,U0,Er(10),Si(10),Hi(10),Gr(10),Gi(10),Kfs,Re,Nrt,Nla
2650 COM Tsr(101),Tsi(101),Acr(10),Aci(10),Qr(101),Qi(101),Zr(10),Zi(10)
2660 Fr=Fi=Sr=Sii=Fs=0
2670 Er=1E-6
2680 CALL Cdivid(PI*X,0,0,1,Ar,Ai)
2690 CALL Csqrt(Ar,Ai,Br,Bi)
2700 FOR S=1 TO Ns
2710 Qr2=Qr(S)*Qr(S)-Qi(S)*Qi(S)
2720 Qi2=2*Qr(S)*Qi(S)
2730 CALL Cexp(Tsi(S)*X,-Tsr(S)*X,Ar,Ai)
2740 CALL Cdivid(Ar,Ai,Tsr(S)-Qr2,Tsi(S)-Qi2,Cr,Ci)
2750 Fr=Fr+Cr
2760 Fi=Fi+Ci
2770 IF S<10 THEN 2830
2780 Err=ABS((Fr-Sr)/Sr)
2790 Eri=ABS((Fi-Sii)/Sii)
2800 IF Err>Er THEN 2830
2810 IF Eri>Er THEN 2830
2820 Fs=1
2830 Sr=Fr
2840 Sii=Fi
2850 IF Fs=1 THEN 2870
2860 NEXT S
2870 Mer=MAX(Err,Eri)
2880 Wr=Br*Sr-Bi*Sii
2890 Wi=Br*Sii+Bi*Sr
2900 Rns=S-1
2910 SUBEND !

2920 SUB Kutta(Ar,Ai,N,Br,Bi,Yntr,Ynti,Yr,Yi,Fn)
2930 RAD
2940 DIM Kr(3),Ki(3)
2950 Hr=(Br-Ar)/N
2960 Hi=(Bi-Ai)/N
2970 Eps=1E-6
2980 Xr=Ar
2990 Xi=Ai
3000 Hhr=Hr/2
3010 Hhi=Hi/2
3020 Ysvr=Yr=Yntr
3030 Ysvi=Yi=Ynti
3040 Xsvr=Xr
3050 Xsvi=Xi
3060 FOR L=0 TO 3
3070 CALL Func(Fn,Ysvr,Ysvi,Xr,Xi,Fr,Fi)
3080 Kr(L)=Hr*Fr-Hi*Fi
3090 Ki(L)=Hr*Fi+Hi*Fr
3100 ON L+1 GOTO 3110,3110,3160,3200
3110 Xr=Xsvr+Hhr
3120 Xi=Xsvi+Hhi
3130 Ysvr=Yr+Kr(L)*.5
3140 Ysvi=Yi+Ki(L)*.5
3150 GOTO 3200
3160 Xr=Xsvr+Hr
3170 Xi=Xsvi+Hi
3180 Ysvr=Yr+Kr(L)
3190 Ysvi=Yi+Ki(L)
```

```
3200 NEXT L
3210 Yr=Yr+(Kr(0)+2*(Kr(1)+Kr(2))+Kr(3))/6
3220 Yi=Yi+(Ki(0)+2*(Ki(1)+Ki(2))+Ki(3))/6
3230 IF Xr>Br-Eps THEN SUBEXIT
3240 Ysvr=Yr
3250 Ysvi=Yi
3260 GOTO 3040
3270 SUBEND !

3280 SUB Func(Fn,Ysvr,Ysvi,Xr,Xi,Fr,Fi)
3290 Ar=Xr*Xr-Xi*Xi
3300 Ai=2*Xr*Xi
3310 IF Fn=2 THEN 3340
3320 CALL Cdivid(1,0,Ysvr-Ar,Ysvi-Ai,Fr,Fi)
3330 SUBEXIT
3340 Br=Ysvr*Ar-Ysvi*Ai
3350 Bi=Ysvr*Ai+Ysvi*Ar
3360 CALL Cdivid(1,0,1-Br,-Bi,Fr,Fi)
3370 SUBEND !

3380 SUB Cpower(N,Xr,Xi,Yr,Yi)
3390 IF N<>0 THEN 3430
3400 Yr=1
3410 Yi=0
3420 SUBEXIT
3430 Yr=Ar=Xr
3440 Yi=Ai=Xi
3450 IF N=-1 THEN 3570
3460 IF N=1 THEN SUBEXIT
3470 FOR L=1 TO ABS(N)-1
3480 Zr=Ar*Xr-Ai*Xi
3490 Zi=Ar*Xi+Ai*Xr
3500 Ar=Zr
3510 Ai=Zi
3520 NEXT L
3530 IF N<0 THEN 3570
3540 Yr=Ar
3550 Yi=Ai
3560 SUBEXIT
3570 CALL Cdivid(1,0,Ar,Ai,Yr,Yi)
3580 SUBEND !

3590 SUB Rt0(X,Y,T0r,T0i)
3600 Ar=Ai=Br=Bi=Cr=Ci=0
3610 CALL Cpower(2,X,Y,Ar,Ai)
3620 CALL Cpower(-1,X,Y,Br,Bi)
3630 Ar=Ar+.5*Br
3640 Ai=Ai+.5*Bi
3650 CALL Cpower(-4,X,Y,Br,Bi)
3660 Ar=Ar+Br/8
3670 Ai=Ai+Bi/8
3680 CALL Cpower(-7,X,Y,Br,Bi)
3690 Ar=Ar+5*Br/32
3700 Ai=Ai+5*Bi/32
3710 CALL Cpower(-10,X,Y,Br,Bi)
3720 Ar=Ar+11*Br/32
3730 Ai=Ai+11*Bi/32
3740 CALL Cpower(-13,X,Y,Br,Bi)
3750 Ar=Ar+539*Br/512
3760 Ai=Ai+539*Bi/512
3770 CALL Cpower(3,X,Y,Cr,Ci)
3780 Cr=-1-Cr*4/3
3790 Ci=-Ci*4/3
3800 CALL Cpower(-3,X,Y,Br,Bi)
3810 Cr=Cr-7*Br/12
3820 Ci=Ci-7*Bi/12
```

```
3830 CALL Cpower(-6,X,Y,Br,Bi)
3840 Cr=Cr-31*Br/48
3850 Ci=Ci-31*Bi/48
3860 CALL Cpower(-9,X,Y,Br,Bi)
3870 Cr=Cr-397*Br/288
3880 Ci=Ci-397*Bi/288
3890 CALL Cexp(Cr,Ci,Br,Bi)
3900 CALL Cpower(2,X,Y,Cr,Ci)
3910 T0r=Ar+2*Br*Ci+2*Bi*Cr
3920 T0i=Ai+2*Bi*Ci-2*Br*Cr
3930 SUBEND !

3940 SUB Azero(Z,R)
3950 R=1+5/(48*Z^2)-5/(36*Z^4)+77125/(82944*Z^6)-108056875/(6967296*Z^8)
3960 R=Z^(2/3)*(R+162375596875/(334430208*Z^10))
3970 SUBEND !

3980 SUB Azerop(Z,R)
3990 R=1-7/(48*Z^2)+35/(288*Z^4)-181223/(207360*Z^6)
4000 R=Z^(2/3)*(R+18683371/(1244160*Z^8)-91145884361/(191102976*Z^10))
4010 SUBEND !

4020 SUB Pseries(X,Qr,Qi,Wr,Wi)
4030 COM W,E0,U0,Er(10),Si(10),Hi(10),Gr(10),Gi(10),Kfs,Re,Nrt,Nla
4040 COM Tsr(101),Tsi(101),Acr(10),Aci(10),Qr(101),Qi(101),Zr(10),Zi(10)
4050 Wr=Wi=0
4060 CALL Cexp(0,PI/4,Ar,Ai)
4070 Br=(Qr*Ar-Qi*Ai)*SQR(X)
4080 Bi=(Qr*Ai+Qi*Ar)*SQR(X)
4090 FOR M=0 TO 10
4100 CALL Cpower(M,Br,Bi,Ar,Ai)
4110 Dr=Acr(M)*Ar-Aci(M)*Ai
4120 Di=Acr(M)*Ai+Aci(M)*Ar
4130 Wr=Wr+Dr
4140 Wi=Wi+Di
4150 NEXT M
4160 SUBEND !

4170 SUB Ac(Qr,Qi)
4180 COM W,E0,U0,Er(10),Si(10),Hi(10),Gr(10),Gi(10),Kfs,Re,Nrt,Nla
4190 COM Tsr(101),Tsi(101),Acr(10),Aci(10),Qr(101),Qi(101),Zr(10),Zi(10)
4200 P2=SQR(PI)
4210 Acr(0)=1
4220 Aci(0)=0
4230 Acr(1)=0
4240 Aci(1)=-P2
4250 Acr(2)=-2
4260 Aci(2)=0
4270 CALL Cpower(-3,Qr,Qi,A3r,A3i)
4280 CALL Cpower(-6,Qr,Qi,A6r,A6i)
4290 CALL Cpower(-9,Qr,Qi,A9r,A9i)
4300 Acr(3)=-(A3i/4)*P2
4310 Aci(3)=(1+A3r/4)*P2
4320 Acr(4)=4/3*(1+A3r/2)
4330 Aci(4)=4/3*A3i/2
4340 Acr(5)=P2/4*(3*A3i/4)
4350 Aci(5)=-(P2/4)*(1+3*A3r/4)
4360 Acr(6)=-(8/15)*(1+A3r+7*A6r/32)
4370 Aci(6)=-(8/15)*(A3i+7*A6i/32)
4380 Acr(7)=-(P2/6)*(5*A3i/4+27*A6i/32)
4390 Aci(7)=P2/6*(1+5*A3r/4+27*A6r/32)
4400 Acr(8)=16/105*(1+3*A3r/2+27*A6r/32)
4410 Aci(8)=16/105*(3*A3i/2+27*A6i/32)
4420 Acr(9)=P2/24*(7*A3i/4+5*A6i/4+21*A9i/64)
4430 Aci(9)=-(P2/24)*(1+7*A3r/4+5*A6r/4+21*A9r/64)
4440 Acr(10)=-(32/945+64*A3r/945+11*A6r/189+7*A9r/270)
```

```
4450 Aci(10)=-(64*A3i/945+11*A6i/189+7*A9i/270)
4460 SUBEND !

4470 SUB Scurv(X,Qr,Qi,Wr,Wi,Sns,Nfp,Mer)
4480 Wr=Wi=0
4490 Ar=Qr*Qr-Qi*Qi
4500 Ai=2*Qr*Qi
4510 Pr=-X*Ai
4520 Pi=X*Ar
4530 Pmag=SQR(Pr*Pr+Pi*Pi)
4540 Psr=Pr*Pr-Pi*Pi
4550 Psi=2*Pr*Pi
4560 IF (Pmag>12) OR (Qr<0) THEN 4600
4570 CALL Fp1(Pr,Pi,Fr,Fi,Sns,Mer)
4580 Nfp=1
4590 GOTO 4620
4600 CALL Fp2(Pr,Pi,Fr,Fi,Sns,Mer)
4610 Nfp=2
4620 CALL Csqrt(Pr*PI,Pi*PI,P2r,P2i)
4630 CALL Cpower(-3,Qr,Qi,A3r,A3i)
4640 CALL Cpower(-6,Qr,Qi,A6r,A6i)
4650 Br=1+P2i-Fr-2*Pr*Fr+2*Pi*Fi
4660 Bi=-(P2r+Fi+2*Pr*Fi+2*Pi*Fr)
4670 Wr=Fr+.25*(A3r*Br-A3i*Bi)
4680 Wi=Fi+.25*(A3r*Bi+A3i*Br)
4690 Br=1-P2r*Pi+P2i-P2i*Pr-2*Pr+5/6*Psr+.5*Psr*Fr-Fr-.5*Psi*Fi
4700 Bi=P2r*Pr-P2r-P2i*Pi-2*Pi+5/6*Psi+.5*Psr*Fi-Fi+.5*Psi*Fr
4710 Wr=Wr+.25*(A6r*Br-A6i*Bi)
4720 Wi=Wi+.25*(A6r*Bi+A6i*Br)
4730 SUBEND !

4740 SUB Fp1(Pr,Pi,Fr,Fi,Sns,Mer)
4750 Cr=Ci=Fr=Fi=Fs=0
4760 Er=1E-4
4770 CALL Csqrt(Pr,Pi,Ar,Ai)
4780 Tr=-Pr
4790 Ti=-Pi
4800 Gr=Br=-Ai
4810 Gi=Bi=Ar
4820 FOR N=1 TO 40
4830 F1=-(1/N)*((2*N-1)/(2*N+1))
4840 Cr=F1*(Br*Tr-Bi*Ti)
4850 Ci=F1*(Br*Ti+Bi*Tr)
4860 Br=Cr
4870 Bi=Ci
4880 Gr=Gr+Br
4890 Gi=Gi+Bi
4900 IF N<3 THEN 4960
4910 Err=ABS((Fr-Gr)/Fr)
4920 Eri=ABS((Fi-Gi)/Fi)
4930 IF Err>Er THEN 4960
4940 IF Eri>Er THEN 4960
4950 Fs=1
4960 Fr=Gr
4970 Fi=Gi
4980 IF Fs=1 THEN 5000
4990 NEXT N
5000 Mer=MAX(Err,Eri)
5010 Sns=N-1
5020 Cr=1-2/SQR(PI)*Fr
5030 Ci=-(2/SQR(PI))*Fi
5040 CALL Cexp(-Pr,-Pi,Br,Bi)
5050 Gr=Br*Cr-Bi*Ci
5060 Gi=Br*Ci+Bi*Cr
5070 CALL Csqrt(Pr*PI,Pi*PI,Ar,Ai)
5080 Fr=1+Ar*Gi+Ai*Gr
```

```
5090 Fi=Ai*Gi-Ar*Gr
5100 SUBEND !

5110 SUB Fp2(Pr,Pi,Fr,Fi,Sns,Mer)
5120 Gr=Gi=Fr=Fi=Fs=0
5130 Er=1E-4
5140 IF ATN(Pi/Pr)<0 THEN 5190
5150 CALL Csqrt(PI*Pr,PI*Pi,Cr,Ci)
5160 CALL Cexp(-Pr,-Pi,Dr,Di)
5170 Gr=2*Dr*Ci+2*Di*Cr
5180 Gi=2*Di*Ci-2*Dr*Cr
5190 CALL Cdivid(1,0,-2*Pr,-2*Pi,Ar,Ai)
5200 Tr=Ar
5210 Ti=Ai
5220 Gr=Gr+Tr
5230 Gi=Gi+Ti
5240 FOR M=2 TO 20
5250 Br=-(2*M-1)*(Ar*Tr-Ai*Ti)
5260 Bi=-(2*M-1)*(Ar*Ti+Ai*Tr)
5270 Tr=Br
5280 Ti=Bi
5290 Gr=Gr+Tr
5300 Gi=Gi+Ti
5310 IF M<4 THEN 5370
5320 Err=ABS((Fr-Gr)/Fr)
5330 Eri=ABS((Fi-Gi)/Fi)
5340 IF Err>Er THEN 5370
5350 IF Eri>Er THEN 5370
5360 Fs=1
5370 Fr=Gr
5380 Fi=Gi
5390 IF Fs=1 THEN 5410
5400 NEXT M
5410 Mer=MAX(Err,Eri)
5420 Sns=M-1
5430 SUBEND
```

## Appendix B

## USEFUL APPROXIMATION FOR THE GROUND-WAVE ATTENUATION

Solutions for the ground-wave attenuation function can be found by using one of three techniques. Two of these are approximations valid at short numerical distances while the third uses a residue series summation that is technically accurate at any range, but which becomes increasingly inefficient (in the number of terms required) as the distance between source and receiver is decreased.

Calculation of the attenuation function using the residue series representation requires knowledge of the poles $t_s$, which are roots of the equation:

$$\frac{dw(t)}{dt} - qw(t) = 0, \tag{B1}$$

where

$$w(t) = \sqrt{\pi}\{Bi(t) - iAi(t),$$

in which $Ai(t)$ and $Bi(t)$ are Airy functions [B1].

An efficient procedure for solution of Eq. (B1) has been described by Hill and Wait [B2]. A summary of this procedure and the regions of validity are presented in Fig. B1. In the first region, denoted by "1," the differential equation

$$\frac{dt_s}{dq} = \frac{1}{t_s - q^2}, \tag{B2}$$

is integrated numerically using a complex, Runge-Kutta technique (Appendix A provides a listing of the numerical techniques employed to solve for the attenuation function) starting from the initial conditions:

$$t_s \mid_{q=0} = t_s(0) = \alpha'_s e^{\left(-\frac{i\pi}{3}\right)}, \tag{B3}$$

in which the set of values $\alpha'_s$ ($s = 1$ to $\infty$) are the real zeroes of $Ai'(-\alpha)$.

In the second region identified by "II" in Fig. B1, the transformation $Q = q^{-1}$ is made to obtain the differential equation:

$$\frac{dt_s}{dQ} = \frac{1}{1 - Q^2 t_s}. \tag{B4}$$

This equation is integrated using the same numerical technique starting from the initial conditions:

$$t_s \mid_{Q=0} = t_s(\infty) = \alpha_s e^{\left(-\frac{i\pi}{3}\right)}, \tag{B5}$$

in which the values $\alpha_s$ ($s = 1$ to $\infty$) are the real zeroes of $Ai(-\alpha)$.

In the region of Fig. B1 in which the ground constants produce $|q| > 1$ and phase of $q > -30°$, an additional root $t_0$ represents a trapped surface wave mode. Depending on the exact value of $q$,




I. $\dfrac{dt_s}{dq} = \dfrac{1}{(t_s - q^2)} \qquad t_s\,|_{q=0} = t_s(0) = \alpha'_s\, e^{\left(-\frac{i\pi}{3}\right)}$

II. $\dfrac{dt_s}{dQ} = \dfrac{1}{(1 - Q^2 t_s)} \qquad t_s\,|_{Q=0} = t_s(\infty) = \alpha_s\, e^{\left(-\frac{i\pi}{3}\right)}$ where $Q = q^{-1}$

III. $t_0 = q^2 + \dfrac{1}{2q} + \dfrac{1}{8q^4} + \dfrac{5}{32q^7} + \dfrac{11}{32q^{10}} + \dfrac{539}{512q^{13}} + \cdots$

IV. $\dfrac{dt_s}{dq} = \dfrac{1}{(t_s - q^2)} \qquad t_0\,|_{q=0} = t_1(0)$

Figure B1 — Methods used for solution of the roots $t_s$ and their regions of validity

two methods are used to obtain this root. The first (given as "III" in Fig. B1) uses the asymptotic expansion [B3]:

$$t_0 \sim q^2 + \frac{1}{2q} + \frac{1}{8q^4} + \frac{5}{32q^7} + \frac{11}{32q^{10}} + \frac{539}{512q^{13}}$$
$$-12q^2 \exp\left(-\frac{4q^3}{3} - 1 - \frac{7}{12q^3} - \frac{31}{48q^6} - \frac{397}{288q^9}\right). \qquad \text{(B6)}$$

The second method (given as "IV" in Fig. B1) uses equation (B2) in the region shown for the solution of the root $t_0$ only, with the initial condition $t_0|_{q=0} = t_1(0)$. All other roots within the applicable region are found using Eq. (B4).

Two approximations are useful for calculation of the ground wave attenuation function at short numerical distances $x$. The first of these is a modification to the flat earth attenuation function for spherical earth and is given [B2] by:

$$W = F(p) + \frac{1}{4q^3}\left[1 - i\sqrt{\pi p} - (1+2p)F(p)\right]$$
$$+ \frac{1}{4q^6}\left[1 - i\sqrt{\pi p}\,(1-p) - 2p + \frac{5p^2}{6} + \left(\frac{p^2}{2} - 1\right)F(p)\right], \qquad \text{(B7)}$$

where

$$F(p) = 1 - i\sqrt{\pi p}\, e^{(-p)} erfc(i\sqrt{p}), \tag{B8}$$

in which $p = ixq^2$ and $erfc$ is the complementary error function [B1].

The approximation is useful at short numerical distance $x < 0.2$ and for values of $|q| > 1$ as shown in Fig. B2. Typically, these conditions can apply to a uniform lossy surface as well as certain layered cases.




Figure B2 — Methods used for calculation of the ground wave and the regions over which they are most suitable

The second short distance approximation is a power series due to Bremmer [B4] given by:

$$W = \sum_{n=0}^{\infty} A_n (e^{\frac{i\pi}{4}} q\sqrt{x})^n , \tag{B9}$$

in which the first five coefficients are:

$$A_0 = 1,$$

$$A_1 = -i\sqrt{\pi},$$

$$A_2 = -2,$$

$$A_3 = i\sqrt{\pi}\left(1 + \frac{1}{4q^3}\right), \text{ and}$$

$$A_4 = \frac{4}{3}\left(1 + \frac{1}{2q^3}\right).$$

Additional terms through $A_{10}$ are available [B2] and, while they are not included here, have been used in the calculations presented in this report. The power series approximation is most useful for a numerical distance $x$ less than approximately 0.2 and for a surface characterized by $|q| < 1$ such as unlayered seawater or thin ice over seawater at low frequencies.

## REFERENCES

[B1] M. Abramowitz and I.A. Stegun, *Handbook of Mathematical Functions*, National Bureau of Standards (Applied Mathematics Series 55, Nov. 1970).

[B2] D.A. Hill and J.R. Wait, ''Ground Wave Attenuation Function for a Spherical Earth with Arbitrary Surface Impedance,'' *Radio Sci.* **15**(3) (1980).

[B3] N.A. Logan, ''Numerical Investigation of Electromagnetic Scattering and Diffraction by Convex Objects,'' Tech. Report AFCRL 66-153, Air Force Geophys. Lab., Air Force Systems Command, Hanscom AFB, Bedford, MA.

[B4] H. Bremmer, ''Applications of Operational Calculus to Ground-Wave Propagation, Particularly for Long Waves,'' *IRE Trans. Antennas Propag.* **AP-6**, 267 (1958).

DEPARTMENT OF THE NAVY

NAVAL RESEARCH LABORATORY
Washington, D.C. 20375-5000